
Programmation GP

Créer un fichier prog.gp avec le contenu suivant :

dist(a,b) = sqrt(a^2+b^2);

et sauvegarder. Sous GP faire (sans retour chariot)

\r

et faire glisser l’icone du fichier dans la fenêtre pour compléter
le nom :

\r .../prog.gp

Essayer la fonction :

? dist(1,2)
%1 = 2.2360679774997896964091736687312762354



Programmation GP

Dans un fichier GP, les sauts de lignes terminent les entrées
sauf si ils sont précédé d’un = d’un \ ou dans une section entre
accolades.

Exemple de fonction

Ajouter dans le fichier prog.gp

fibo(n)=
{
my(u0=0,u1=1);
for(i=2,n,
[u0,u1]=[u1,u0+u1]);

u1;
}

et essayer

? \r
? fibo(100)

I Mettre les accolades sur la ligne après le signe =.
I Terminer la fonction par un point-virgule.

fibo(n)=
{
...;

}

I Declarer les variables locales avec my().
I Ne pas déclarer l’indice de boucle qui est local à la boucle.

...
my(u0=0,u1=1);

...

I La fonction retourne le dernier resultat calculé.
I Indenter le code pour la lisibilité.

fibo(n)=
{
....
u1;

}

I Indenter le code pour la lisibilité.

fibo(n)=
{
__my(u0=0,u1=1);
__for(i=2,n,
____[u0,u1]=[u1,u0+u1]);
__u1;
}

Boucle while
Ajouter

rho(n)=
{
my(x=2,y=5);
while(gcd(y-x,n)==1,

x=(x^2+1)%n;
y=(y^2+1)%n; y=(y^2+1)%n
);

gcd(n,y-x);
}

et faire

\r
rho(2^64+1)
%1 = 274177

Contrôle de flux
wieferich(n)=
{
forprime(p=2,n,
if(Mod(2,p^2)^(p-1)==1,
return(p)));

}
wieferich2(n)=
{
my(r);
forprime(p=2,n,
if(Mod(2,p^2)^(p-1)==1,r=p;break));

r;
}
? wieferich(10000)
%4 = 1093

Constructeurs

? V=vector(10,i,1/i)
%1 = [1,1/2,1/3,1/4,1/5,1/6,1/7,1/8,1/9,1/10]
? [1/i|i<-[1..10]]
%2 = [1,1/2,1/3,1/4,1/5,1/6,1/7,1/8,1/9,1/10]
? M=matrix(4,4,i,j,i*j)
%3 = [1,2,3,4;2,4,6,8;3,6,9,12;4,8,12,16]

Les variable i et j sont locales et ne doivent pas être déclarée.

forvec

Au lieu de

s3(n)=
{
my(m=sqrtint(n));
for (i=1,m,
for (j=1,m,
for (k=1,m,
if (i^2+j^2+k^2==n,

return([i,j,k])))));
}
? s3(12345)
%2 = [4,77,80]

forvec

Avec forvec :

s3(n)=
{
my(m=sqrtint(n));
forvec(v=vector(3,i,[1,m]),
if (v*v~==n,
return(v)));

}
? s3(12345)
%2 = [4,77,80]

Pour un meilleur algorithme, utiliser qfsolve.

tableaux associatifs

birthday(n)=
{
my(M = Map());
for(i=1,oo,
my(x=random(n), j);
if(mapisdefined(M,x,&j),
return([i,j]));

mapput(~M,x,i));
}
? birthday(2^30)
%2 = [417,383]


